



Chemical Engineering

UNIVERSITI TEKNOLOGI PETRONAS

CDB 4313 – HEAT INTEGRATION

AUTOMATED HEAT EXCHANGER NETWORK (HEN) DESIGN

ZULFAN ADI PUTRA, PDENG

OFFICE: 05-03-08
TELP: 05 368 7562
EMAIL: zulfan.adiputra@utp.edu.my

Discussion time: Friday 15.00 – 17.00

COURSE OVERVIEW



Chemical Engineering
Inspiring Potential·Generating Futures

AUTOMATED HEAT EXCHANGER DESIGN

Two well-known models:

1. Transhipment model

2. Superstructure model

Transhipment model

TEMPERATURE INTERVAL



Temperature Interval



Heat is only transferred from a higher level to a lower level

Notations:

i = {U, H1, H2};
k = {C1, C2};
J = {J0, J1, …, Jn}

ENERGY BALANCE IN COLD STREAMS

For every cold stream ($\forall\ k$) in every interval ($\forall\ J$), it can only take a specified amount of energy ($P_{k,J}$)

Interval J0

$$s_{U,C1,J0} + s_{H1,C1,J0} + s_{H2,C1,J0} = P_{C1,J0}$$
$$s_{U,C2,J0} + s_{H1,C2,J0} + s_{H2,C2,J0} = P_{C2,J0}$$

Interval J1

$$s_{U,C1,J1} + s_{H1,C1,J1} + s_{H2,C1,J2} = P_{C1,J2}$$
$$s_{U,C2,J1} + s_{H1,C2,J1} + s_{H2,C2,J2} = P_{C2,J2}$$

Notations:

i = {U, H1, H2};
k = {C1, C2};
J = {J0, J1, …, Jn}

$$\sum_{i} s_{i,k,J} = P_{k,J} \quad \forall\ k, J$$

```
CSBAL(K,J).. SUM(I, S(I,K,J)) =E= P(K,J) ;
```

Chemical Engineering
Inspiring Potential · Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

Energy balance in hot streams

For every hot stream ($\forall\ i$) in every interval ($\forall\ J$), it can only supply a specified amount of energy ($R_{i,J}$)

Interval J0

$$D_{U,J0} + s_{U,C1,J0} + s_{U,C2,J0} = R_{U,J0}$$
$$D_{H1,J0} + s_{H1,C1,J0} + s_{H1,C2,J0} = R_{H1,J0}$$
$$D_{H2,J0} + s_{H2,C1,J0} + s_{H2,C2,J0} = R_{H2,J0}$$
$$D_{i,J0} + \sum_{k} s_{i,k,J0} = R_{i,J0} \quad \forall\ i\ in\ J_0$$

Notations:

i = {U, H1, H2};
k = {C1, C2};
J = {J0, J1, …, Jn}

Interval J1

$$D_{U,J1} - D_{U,J0} + s_{U,C1,J1} + s_{U,C2,J1} = R_{U,J1}$$
$$D_{H1,J1} - D_{H1,J0} + s_{H1,C1,J1} + s_{H1,C2,J1} = R_{H1,J1}$$
$$D_{H2,J1} - D_{H2,J0} + s_{H2,C1,J1} + s_{H2,C2,J1} = R_{H2,J1}$$
$$D_{i,J} - D_{U,(J-1)} + \sum_{k} s_{i,k,J} = R_{i,J} \quad \forall\ i, J\ except\ in\ J_0$$

HEAT EXCHANGING MATCHES, $Y_{I,K}$

For every hot and cold stream, a heat exchanging match,
$Y_{i,k}$ does exist (1) if there is at least one $s_{i,k,J}$ in any interval
$Y_{i,k}$ does not exist (0) if there is none of $s_{i,k,J}$ in any interval

$$s_{U,C1,J0} + s_{U,C1,J1} + \cdots + s_{U,C1,J} - \gamma \cdot Y_{U,C1} \leq 0$$
$$s_{U,C2,J0} + s_{U,C2,J1} + \cdots + s_{U,C2,J} - \gamma \cdot Y_{U,C2} \leq 0$$

$$\sum_{J} s_{i,k,J} - \gamma \cdot Y_{i,k} \leq 0 \quad \forall\, i, k$$

Notations:

i = {U, H1, H2};
k = {C1, C2};
J = {J0, J1, …, Jn}

```
HTINEQ1(I,K) .. SUM(J, S(I,K,J))-GAMMA*Y(I,K) =L= 0 ;
```

OBJECTIVE FUNCTION

The objective function is to minimize the number of heat exchangers
This means also to minimize the number of matches

$$Minimize\ z$$

$$z = Y_{U,C1} + Y_{U,C2} + \ \dots + Y_{i,k}$$

$$z = \sum_{i}\sum_{k} Y_{i,k}$$

Notations:

i = {U, H1, H2};
k = {C1, C2};
J = {J0, J1, …, Jn}

```
MINMATCH .. Z =E= SUM((I,K), Y(I,K));
```

Chemical Engineering
Inspiring Potential · Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

GAMS MODEL – DESIGN ABOVE PINCH

```
SETS
I hot streams above pinch        / U, H1,H2 /
K cold streams above pinch       / C1,C2 /
J temperature intervals          / J0*J3 / ;

SCALAR GAMMA /10000/;

TABLE R(I,J) load of hot stream I1 in interval K
         J0      J1      J2      J3
U        7.5     0       0       0
H1       1.5     6       1.5     6
H2       0       0       2.5     10;

TABLE P(K,J) load of cold stream K1 in interval J
         J0      J1      J2      J3
C1       0       0       0       8
C2       0       12      3       12 ;

VARIABLES
S(I,K,J) heat exchanged hot and cold streams
D(I,J)   heat of hot streams flowing between intervals
Y(I,K)   existence of match
Z        total number of matches ;

POSITIVE VARIABLE S
POSITIVE VARIABLE D
BINARY VARIABLE Y ;

EQUATIONS
MINMATCH          objective function-number of matches
HSBAL1(I,J)       heat balances of hot stream I in INTERVAL J ne 1
HSBAL(I,J)        heat balances of hot stream I in INTERVAL 1
CSBAL(K,J)        heat balances of cold stream J1 in K
HTINEQ1(I,K)      heat transferred inequalities;

MINMATCH .. Z =E= SUM((I,K), Y(I,K));
HSBAL1(I,J)$(ORD(J) NE 1) .. D(I,J)-D(I,J-1)+ SUM(K,S(I,K,J)) =E= R(I,J);
HSBAL(I,J)$(ORD(J) EQ 1) .. D(I,J)+SUM(K,S(I,K,J)) =E= R(I,J);
CSBAL(K,J).. SUM(I, S(I,K,J)) =E= P(K,J) ;
HTINEQ1(I,K) .. SUM(J, S(I,K,J))-GAMMA*Y(I,K) =L= 0 ;

MODEL TSHIP /ALL/ ;
SOLVE TSHIP USING MIP MINIMIZING Z;
DISPLAY S.L, D.L, Y.L,Z.l;
```

GAMS RESULT

```
GAMS Rev 237  WIN-VS8 23.7.3 x86/MS Windows          02/23/17 13:08:04 Page 1
G e n e r a l   A l g e b r a i c   M o d e l i n g   S y s t e m
C o m p i l a t i o n


SETS
I hot streams above pinch        / U, H1, H2/
K cold streams above pinch       / C1, C2/
J temperature intervals          / J0*J3/;

SCALAR
GAMMA / 10000/;

TABLE R(I,J)  load of hot stream I in interval J
          J0       J1       J2       J3
U         7.5      0        0        0
H1        1.5      6        1.5      6
H2        0        0        2.5      10;

TABLE P(K,J)  load of cold stream K in interval J
          J0       J1       J2       J3
C1        0        0        0        8
C2        0        12       3        12;

VARIABLES
S(I,K,J)  heat exchanged hot and cold streams
D(I,J)    heat of hot streams flowing between intervals
Y(I,K)    existence of match
Z         total number of matches;

POSITIVE VARIABLE S
POSITIVE VARIABLE D
BINARY VARIABLE Y;

EQUATIONS
MINMATCH          objective function-number of matches
HSBAL1(I,J)       heat balances of hot stream I in interval J for interval
not equal to 1
```





Chemical Engineering

Inspiring Potential · Generating Futures

GAMS RESULT

UNIVERSITI TEKNOLOGI PETRONAS

```
 34  HSBAL(I,J)        heat balances of hot stream I in interval 1
 35  CSBAL(K,J)        heat balances of cold stream J in K
 36  HTINEQ(I,K)       heat transferred inequalities;
 37
 38  MINMATCH .. Z =E= SUM((I,K), Y(I,K));
 39  HSBAL1(I,J)$(ORD(J) NE 1).. D(I,J)-D(I,J-1)+SUM(K,S(I,K,J)) =E= R(I,J);
 40  HSBAL(I,J)$(ORD(J) EQ 1).. D(I,J)+SUM(K,S(I,K,J)) =E= R(I,J);
 41  CSBAL(K,J).. SUM(I,s(I,K,J)) =E= P(K,J);
 42  HTINEQ(I,K).. SUM(J,S(I,K,J))-GAMMA*Y(I,K) =L= 0;
 43
 44  MODEL TSHIP /ALL/;
 45  SOLVE TSHIP USING MIP MINIMIZING Z;
 46  DISPLAY S.l, D.l, Y.l, Z.l;


COMPILATION TIME     =        0.000 SECONDS      3 Mb  WIN237-237 Aug 23, 2011
GAMS Rev 237  WIN-VS8 23.7.3 x86/MS Windows             02/23/17 13:09:30 Page 2
G e n e r a l   A l g e b r a i c   M o d e l i n g   S y s t e m
Equation Listing    SOLVE TSHIP Using MIP From line 45


---- MINMATCH  =E=  objective function-number of matches

MINMATCH..  - Y(U,C1) - Y(U,C2) - Y(H1,C1) - Y(H1,C2) - Y(H2,C1) - Y(H2,C2) + Z
      =E= 0 ; (LHS = 0)


---- HSBAL1  =E=  heat balances of hot stream I in interval J for interval not e
                  qual to 1

HSBAL1(U,J1)..  S(U,C1,J1) + S(U,C2,J1) - D(U,J0) + D(U,J1) =E= 0 ; (LHS = 0)

HSBAL1(U,J2)..  S(U,C1,J2) + S(U,C2,J2) - D(U,J1) + D(U,J2) =E= 0 ; (LHS = 0)

HSBAL1(U,J3)..  S(U,C1,J3) + S(U,C2,J3) - D(U,J2) + D(U,J3) =E= 0 ; (LHS = 0)

REMAINING 6 ENTRIES SKIPPED
```

GAMS RESULT

```
---- HSBAL  =E=  heat balances of hot stream I in interval 1

HSBAL(U,J0)..  S(U,C1,J0) + S(U,C2,J0) + D(U,J0) =E= 7.5 ;

      (LHS = 0, INFES = 7.5 ****)

HSBAL(H1,J0)..  S(H1,C1,J0) + S(H1,C2,J0) + D(H1,J0) =E= 1.5 ;

      (LHS = 0, INFES = 1.5 ****)

HSBAL(H2,J0)..  S(H2,C1,J0) + S(H2,C2,J0) + D(H2,J0) =E= 0 ; (LHS = 0)


---- CSBAL  =E=  heat balances of cold stream J in K

CSBAL(C1,J0)..  S(U,C1,J0) + S(H1,C1,J0) + S(H2,C1,J0) =E= 0 ; (LHS = 0)

CSBAL(C1,J1)..  S(U,C1,J1) + S(H1,C1,J1) + S(H2,C1,J1) =E= 0 ; (LHS = 0)

CSBAL(C1,J2)..  S(U,C1,J2) + S(H1,C1,J2) + S(H2,C1,J2) =E= 0 ; (LHS = 0)

REMAINING 5 ENTRIES SKIPPED


---- HTINEQ  =L=  heat transferred inequalities

HTINEQ(U,C1)..  S(U,C1,J0) + S(U,C1,J1) + S(U,C1,J2) + S(U,C1,J3)

      - 10000*Y(U,C1) =L= 0 ; (LHS = 0)

HTINEQ(U,C2)..  S(U,C2,J0) + S(U,C2,J1) + S(U,C2,J2) + S(U,C2,J3)

      - 10000*Y(U,C2) =L= 0 ; (LHS = 0)

HTINEQ(H1,C1)..  S(H1,C1,J0) + S(H1,C1,J1) + S(H1,C1,J2) + S(H1,C1,J3)

      - 10000*Y(H1,C1) =L= 0 ; (LHS = 0)
```

Chemical Engineering
Inspiring Potential · Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

GAMS RESULT

```
GAMS Rev 237  WIN-VS8 23.7.3 x86/MS Windows            02/23/17 13:09:30 Page 4
G e n e r a l   A l g e b r a i c   M o d e l i n g   S y s t e m
Model Statistics    SOLVE TSHIP Using MIP From line 45


MODEL STATISTICS

BLOCKS OF EQUATIONS           5     SINGLE EQUATIONS           27
BLOCKS OF VARIABLES           4     SINGLE VARIABLES           43
NON ZERO ELEMENTS           106     DISCRETE VARIABLES          6


GENERATION TIME      =        0.327 SECONDS      4 Mb  WIN237-237 Aug 23, 2011


EXECUTION TIME       =        0.327 SECONDS      4 Mb  WIN237-237 Aug 23, 2011
GAMS Rev 237  WIN-VS8 23.7.3 x86/MS Windows            02/23/17 13:09:30 Page 5
G e n e r a l   A l g e b r a i c   M o d e l i n g   S y s t e m
Solution Report     SOLVE TSHIP Using MIP From line 45


               S O L V E      S U M M A R Y

     MODEL   TSHIP               OBJECTIVE  Z
     TYPE    MIP                 DIRECTION  MINIMIZE
     SOLVER  CPLEX               FROM LINE  45

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE               4.0000

 RESOURCE USAGE, LIMIT          0.296      1000.000
 ITERATION COUNT, LIMIT         4     2000000000

IBM ILOG CPLEX   Jul 14, 2011 23.7.3 WIN 27723.27726 VS8 x86/MS Windows
Cplex 12.3.0.0

MIP status(101): integer optimal solution
Fixing integer variables, and solving final LP...
```

UNIVERSITI TEKNOLOGI PETRONAS

GAMS RESULT

```
GAMS Rev 237  WIN-VS8 23.7.3 x86/MS Windows           02/23/17 13:09:30 Page 6
G e n e r a l   A l g e b r a i c   M o d e l i n g   S y s t e m
E x e c u t i o n


----     46 VARIABLE S.L  heat exchanged hot and cold streams

               J1          J2          J3

U .C2        5.000       0.500       2.000
H1.C1                                8.000
H1.C2        7.000
H2.C2                    2.500      10.000


----     46 VARIABLE D.L  heat of hot streams flowing between intervals

            J0          J1          J2

U        7.500       2.500       2.000
H1       1.500       0.500       2.000


----     46 VARIABLE Y.L  existence of match

            C1          C2

U                    1.000
H1       1.000       1.000
H2                   1.000


----     46 VARIABLE Z.L                    =        4.000  total number of match
                                                            es


EXECUTION TIME       =        0.015 SECONDS      3 Mb  WIN237-237 Aug 23, 2011
```



GAMS RESULT



Not so handy ☹
Check the temperatures and simplify

GAMS RESULT - SIMPLIFY



Chemical Engineering
Inspiring Potential · Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

Superstructure model

SUPERSTRUCTURE



Superstructure with 2 stages

GAMS CODE


```
Sets i hot streams /1*2/
     j cold streams /1*2/;

Scalar nok  number of stages in superstructure / 2 /;

Set   k          temperature locations  nok + 1 /1*3/
      st(k)      stages
      first(k)   first temperature location
      last(k)    last temperature location ;


st(k)     = yes$(ord(k) lt card(k))   ;
first(k)  = yes$(ord(k) eq 1)         ;
last(k)   = yes$(ord(k) eq card(k))   ;
```

GAMS CODE

```
Parameters

fh(i)        heat capacity flowrate of hot stream
fc(j)        heat capacity flowrate of cold stream
thin(i)      supply temp. of hot stream
thout(i)     target temp. of hot stream
tcin(j)      supply temp. of cold stream
tcout(j)     target temp. of cold stream
ech(i)       heat content hot i
ecc(j)       heat content cold j
hh(i)        stream-individual film coefficient hot i,
hc(j)        stream-individual film coefficient cold j,
hucost       cost of heating utility,
cucost       cost of cooling utility,
unitc        fixed charge for exchanger,
acoeff       area cost coefficient for exchangers,
hucoeff      area cost coefficient for heaters,
cucoeff      area cost coefficient for coolers,
aexp         cost exponent for exchangers,
hhu          stream-individual film coefficient hot utility,
hcu          stream-individual film coefficient cold utility,
thuin        inlet temperature hot utility,
thuout       outlet temperature hot utility,
tcuin        inlet temperature cold utility,
tcuout       outlet temperature cold utility,
gamma(i,j)   upper bound of driving force,
a(i,j,k)     area for exchanger for match ij in interval k (chen approx.),
al(i,j,k)    area calculated with log mean,
acu(i)       area coolers,
ahu(j)       area heaters,
tmapp        minimum approach temperature
costheat     cost of heating,
costcool     cost of cooling,
invcost      investment cost   ;
```

GAMS CODE

```
Binary Variables      z(i,j,k), zcu(i), zhu(j)  ;

Positive Variables
  th(i,k)     temperature of  hot stream i as it enters stage k
  tc(j,k)     temperature of cold stream j as it leaves stage k
  q(i,j,k)    energy exchanged between i and j in stage k
  qc(i)       energy exchanged between i and the cold utility
  qh(j)       energy exchanged between j and the hot utility
  dt(i,j,k)   approach between i and j at location k
  dtcu(i)     approach between i and the cold utility
  dthu(j)     approach between j and the hot utility    ;

Variable cost    hen and utility cost       ;

Equations
 eh(i,k)        energy exchanged by hot  stream i in stage k
 eqc(i,k)       energy exchanged by hot  stream i with the cold utility
 teh(i)         total energy exchanged by hot  stream i
 ec(j,k)        energy exchanged by cold stream j in stage k
 eqh(j,k)       energy exchanged by cold stream j with the hot utility
 tec(j)         total energy exchanged by cold  stream j
 month(i,k)     monotonicity of th
 montc(j,k)     monotonicity of tc
 monthl(i,k)    monotonicity of th k = last
 montcf(j,k)    monotonicity of tc for k = 1
 tinh(i,k)      supply temperature of hot streams
 tinc(j,k)      supply temperature of cold streams
 logq(i,j,k)    logical constraints on  q
 logqh(j)       logical constraints on qh(j)
 logqc(i)       logical constraints on qc(i)
 logdth(i,j,k)  logical constraints on dt at the  hot end
 logdtc(i,j,k)  logical constraints on dt at the cold end
 logdtcu(i,k)   logical constraints on dtcu
 logdthu(j,k)   logical constraints on dthu
 obj            objective function         ;
```

GAMS CODE

```
teh(i)..  (thin(i)-thout(i))*fh(i)  =e= sum((j,st), q(i,j,st)) + qc(i)  ;
tec(j)..  (tcout(j)-tcin(j))*fc(j)  =e= sum((i,st), q(i,j,st)) + qh(j)  ;

eh(i,k)$st(k).. fh(i)*(th(i,k) - th(i,k+1)) =e= sum(j, q(i,j,k))   ;
ec(j,k)$st(k).. fc(j)*(tc(j,k) - tc(j,k+1)) =e= sum(i,q(i,j,k)) ;

eqc(i,k)$last(k)..  fh(i)*(th(i,k) - thout(i)) =e= qc(i) ;
eqh(j,k)$first(k).. fc(j)*(tcout(j) - tc(j,k)) =e= qh(j) ;

tinh(i,k)$first(k).. thin(i) =e= th(i,k) ;
tinc(j,k)$last(k)..  tcin(j) =e= tc(j,k) ;

month(i,k)$st(k).. th(i,k) =g= th(i,k+1) ;
montc(j,k)$st(k).. tc(j,k) =g= tc(j,k+1) ;

monthl(i,k)$last(k).. th(i,k) =g= thout(i) ;
montcf(j,k)$first(k)..tcout(j) =g= tc(j,k) ;

logq(i,j,k)$st(k)..q(i,j,k) - min(ech(i), ecc(j))*z(i,j,k) =l= 0 ;

logqc(i)..qc(i) - ech(i)*zcu(i) =l= 0 ;
logqh(j)..qh(j) - ecc(j)*zhu(j) =l= 0 ;

logdth(i,j,k)$st(k)..dt(i,j,k) =l= th(i,k) - tc(j,k) +
                                   gamma(i,j)*(1 - z(i,j,k)) ;

logdtc(i,j,k)$st(k)..dt(i,j,k+1) =l= th(i,k+1)-tc(j,k+1) +
                                     gamma(i,j)*(1 - z(i,j,k)) ;

logdthu(j,k)$first(k)..dthu(j) =l= (thuout - tc(j,k)) ;
logdtcu(i,k)$last(k)..dtcu(i) =l= th(i,k) - tcuout ;
```

GAMS CODE

```
obj..cost =e= unitc*(sum((i,j,st),z(i,j,st)) +
              sum(i,zcu(i)) + sum(j,zhu(j))) +

              acoeff*sum((i,j,k),(q(i,j,k)*((1/hh(i))+(1/hc(j))))/
              (((dt(i,j,k)*dt(i,j,k+1)*(dt(i,j,k) + dt(i,j,k+1))/2 +
              1e-6)**0.33333) + 1e-6) + 1e-6)**aexp) +

              hucoeff*(sum(j,(qh(j)*((1/hc(j))+1/hhu))/
              (((thuin-tcout(j))*dthu(j)*((thuin-tcout(j)+dthu(j))/2)+
              1e-6)**0.33333) + 1e-6)**aexp) +

              cucoeff*sum(i,(qc(i)*((1/hh(i))+(1/hcu))/
              (((thout(i)-tcuin)*dtcu(i)*(thout(i)-tcuin+dtcu(i))/2 +
              1e-6)**0.33333) + 1e-6)**aexp) +

              sum(j,qh(j)*hucost) + sum(i,qc(i)*cucost) ;


* process streams

* hot

 thin('1')=650;  thout('1')=370;  fh('1')=10; hh('1')=1;
 thin('2')=590;  thout('2')=370;  fh('2')=20; hh('2')=1;

* cold

  tcin('1')=410;  tcout('1')=650;  fc('1')=15;  hc('1')=1;
  tcin('2')=350;  tcout('2')=500;  fc('2')=13;  hc('2')=1;

* costs and coefficients

 hucost =80; hucoeff =150; thuin  =680; thuout =680; hhu   =5;
 cucost =15; cucoeff =150; tcuin  =300; tcuout =320; hcu   =1;
```

Gams Code

```
unitc  =5500;   acoeff =150;  aexp   =1;


tmapp = 10;

* bounds

dt.lo(i,j,k) = tmapp ;
dthu.lo(j) = tmapp ;
dtcu.lo(i) = tmapp ;

th.up(i,k) = thin(i) ;
th.lo(i,k) = thout(i) ;
tc.up(j,k) = tcout(j) ;
tc.lo(j,k) = tcin(j) ;

* initialization

th.l(i,k) = thin(i) ;
tc.l(j,k) = tcin(j) ;

dthu.l(j) = thuout - tcin(j) ;
dtcu.l(i) = thin(i) - tcuout ;

ech(i) = fh(i)*(thin(i) - thout(i)) ;
ecc(j) = fc(j)*(tcout(j) - tcin(j)) ;

gamma(i,j) = max(0,tcin(j) - thin(i), tcin(j) - thout(i),
                 tcout(j) - thin(i), tcout(j) - thout(i)) ;

dt.l(i,j,k) = thin(i) - tcin(j) ;

q.l(i,j,k)$st(k) = min(ech(i),ecc(j)) ;
```

Chemical Engineering
Inspiring Potential·Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

GAMS CODE

```
Model super/all/ ;

Option optcr    = 0 ;
Option limrow   = 0 ;
Option limcol   = 0 ;
Option solprint = off ;
Option sysout   = off ;


Solve super using minlp minimizing cost ;

*  areas by chen approximation

a(i,j,k)$st(k) = q.l(i,j,k)*((1/hh(i))+(1/hc(j)))/
                 (2/3*sqrt(dt.l(i,j,k)*dt.l(i,j,k+1)) +
                  1/6*(1e-8 + dt.l(i,j,k) + dt.l(i,j,k+1))) ;

*  areas by log mean temperature

al(i,j,k)$st(k) = (q.l(i,j,k)*((1/hh(i))+(1/hc(j)))))/
                  (dt.l(i,j,k)*dt.l(i,j,k+1)*
                  (dt.l(i,j,k)+dt.l(i,j,k+1))/2)**0.33333  ;


display a,al ;


* areas of heaters and coolers

ahu(j)  =  (qh.l(j)*((1/hc(j))  + (1/hhu))/(((thuin-tcout(j))*dthu.l(j)*
           ((thuin-tcout(j)+dthu.l(j)))/2) + 1e-6)**0.33333)  ;

acu(i)  =  (qc.l(i)*((1/hh(i))+(1/hcu))/(((thout(i)-tcuin)*dtcu.l(i)*
           (thout(i)-tcuin+dtcu.l(i))/2 + 1e-6)**0.33333)) ;
```

GAMS CODE

```
*  areas by log mean temperature

al(i,j,k)$st(k) = (q.l(i,j,k)*((1/hh(i))+(1/hc(j))))/
                  (dt.l(i,j,k)*dt.l(i,j,k+1)*
                  (dt.l(i,j,k)+dt.l(i,j,k+1))/2)**0.33333  ;


display a,al ;


* areas of heaters and coolers

ahu(j) = (qh.l(j)*((1/hc(j)) + (1/hhu))/(((thuin-tcout(j))*dthu.l(j)*
         ((thuin-tcout(j)+dthu.l(j)))/2) + 1e-6)**0.33333)  ;

acu(i) = (qc.l(i)*((1/hh(i))+(1/hcu))/(((thout(i)-tcuin)*dtcu.l(i)*
         (thout(i)-tcuin+dtcu.l(i))/2 + 1e-6)**0.33333)) ;


display acu, ahu ;


*  utility costs

costheat = sum(j,qh.l(j)*hucost) ;
costcool = sum(i,qc.l(i)*cucost) ;


display costheat, costcool ;


*  investment cost

invcost = cost.l - costheat - costcool ;


display invcost ;
```

Chemical Engineering
Inspiring Potential·Generating Futures

Explanation on the code (1)



$$ech\ (1) = fh(1)[thout\ (1) - thin\ (1)]$$

$$ech\ (2) = fh(2)[thout\ (2) - thin\ (2)]$$

$$ecc\ (1) = fc(1)[tcout\ (1) - tcin\ (1)]$$

$$ecc\ (2) = fc(2)[tcout\ (2) - tcin\ (2)]$$

Chemical Engineering

Inspiring Potential · Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

EXPLANATION ON THE CODE (2)



$$mCp_{(i)}\left(T_{in}(i) - T_{out}(i)\right) = \sum_{st}\sum_{j} q(i,j,st)\ \forall i + q_c(i)$$

$$teh(1): [thin(1) - thwt(1)]\ fh(1) = \underbrace{q(1,1,1) + q(1,2,1)}_{\sum_i q(i,1,1)} + \underbrace{q(1,1,2) + q(1,2,2)}_{\sum_i q(i,1,2)} + q_c(1)$$

$$\sum_{st}\sum_{j} q(1,j,st) \qquad + q_c(1)$$

$$teh(2): thin(2) - thout(2)]\ fh(2) = \sum_{st}\sum_{j} q(s,j,st) \qquad + q_c(2)$$

Chemical Engineering
Inspiring Potential·Generating Futures

EXPLANATION ON THE CODE (3)



$$mCp_{(j)}\left(T_{out}(j) - T_{in}(j)\right) = \sum_{st}\sum_{i} q(i,j,st)\ \forall j + q_h(j)$$

$$tec(1): [tcout\ (1) - tcin(1)]\ fc(1) = \underbrace{\underbrace{q(1{,}1{,}1) + q(2{,}1{,}1)}_{\sum_i q(i,1,1)} + \underbrace{q(1{,}1{,}2) + q(2{,}1{,}2)}_{\sum_i q(i,1,2)}}_{\sum_{st}\sum_i q(i,1,st)} + q_h(1)$$

$$tec(2): [tcout\ (2) - tcin(2)]\ fc\ (2) = \sum_{st}\sum_{i} q(i,2,st) \quad + q_h(2)$$

EXPLANATION ON THE CODE (4)

$$eh(1,1)\ stage\ 1_{k=1}: fh\ (1)[\ th\ (1,1) - th\ (1,2)] = \underbrace{q(1,1,1) + q(1,2,1)}_{\sum_j q(1,j,1)}$$

$$eh(1,2)\ stage\ 2_{k=2}: fh\ (1)[\ th\ (1,2) - th\ (1,3)] = \sum_j q(1,j,2)$$

Only two stages

$$eh\ (2,1)\ stage\ 1: fh\ (2)[th\ (2,1) - th(2,2)] = \sum_j q(2,j,1)$$

$$eh\ (2,2)\ stage\ 2: fh\ (2)[th\ (2,2) - th(2,3)] = \sum_j q(2,j,2)$$

$$ec\ (1,1)\ stage\ 1: fc\ (1)[tc\ (1,1) - tc(1,2)] = \underbrace{q\ (1,1,1) + q(2,1,1)}_{\sum_i q(i,1,1)}$$

$$ec\ (1,2)\ stage\ 2: fc\ (1)[tc\ (1,2) - tc(1,3)] = \sum_i q(i,1,2)$$

$$mCp_i = (T_{i,k} - T_{i,k+1}) = \sum_j q(i,j,k) \quad \forall i,k$$

$$mCp_i = (T_{i,k} - T_{i,k+1}) = \sum_j q(i,j,k) \quad \forall i,k$$

$$ec\ (2,1)\ stage\ 1: fc\ (2)[tc\ (2,1) - tc(2,2)] = \sum_i q(i,2,1)$$

$$eh\ (2,2)\ stage\ 2: fh\ (2)[tc\ (2,2) - tc(2,3)] = \sum_i q(i,2,2)$$

$$mCp_i = (T_{i,k} - T_{i,k+1}) = \sum_j q(i,j,k) \quad \forall i,k$$

$$mCp_i = (T_{i,k} - T_{i,k+1}) = \sum_i q(i,j,k) \quad \forall i,k$$

EXPLANATION ON THE CODE (2)



$$\underbrace{eq_c\ (i,k)}_{eq_c(1,3)}\ @\ cost\ interval:\ fh\ (1)[th\ (1{,}3) - thout\ (1)] = q_c(1)$$

$$eqh\ (1{,}1)@\ first\ int\!: fc(1)[tcout\ (1) - tc(1{,}1)] = qh\ (1)$$

$$eqh\ (1{,}1)\!: fc(2)[tcout\ (2) - tc(2{,}1)] = qh\ (2)$$

$$tinh\ (1{,}1)@\ first\ int\!: thin\ (1) = th\ (1{,}1)$$

$$tinh\ (2{,}1)\!: thin\ (2) = th\ (2{,}1)$$

$$tinc\ (1{,}3)@\ last\ int\!: tcin\ (1) = tc\ (1{,}3)$$

$$tinc\ (2{,}3)\!: tcin\ (2) = tc\ (2{,}3)$$

EXPLANATION ON THE CODE (2)



$month\ (1{,}1)@\ stage\ 1{:}\ th(1{,}1) \gg th(1{,}2)$

$month\ (1{,}2)@\ stage\ 2{:}\ th(1{,}2) \gg th(1{,}3)$

$\}\ T_{in}\ i,k \gg T_{in}\ i,k+1\ \forall k,i$

$montc\ (1{,}1)@\ stage\ 1{:}\ tc(1{,}1) \gg tc(1{,}2)$

$montc\ (1{,}2)@\ stage\ 2{:}\ tc(1{,}2) \gg tc(1{,}3)$

$\}\ T_{out}\ j,k \gg T_{out}\ j,k+1\ \forall k,j$

$monthl\ (1{,}3)@\ last\ int{:}\ th(1{,}3) \gg thout\ (1)$

$monthl\ (2{,}3){:}\ th(2{,}3) \gg thout\ (2)$

} *It may need hot utility*

$T_{i,3} \gg T_{out}\, i, \forall j$

$montcf\ (1{,}1)@\ first\ int{:}\ tcout(1) \gg tc(1{,}1)$

$montcf\ (2{,}1){:}\ tcout(2) \gg tc(2{,}1)$

$\}\ T_{out} j \gg Tj, \forall j$

Chemical Engineering
Inspiring Potential·Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

EXPLANATION ON THE CODE (2)



$\log_q(i,j,k)$ @ *stages: Logical constraints*

$\log_q(1,1,1)$ @ *stage* 1: $q(1,1,1) - \min(ech(1), ecc(1))\, z_{1,1,1} \ll 0$

$\log_q(1,1,2)$ @ *stage* 2: $q(1,1,2) - \min(ech(1), ecc(1))\, z_{1,1,2} \ll 0$

$\log_q(1,2,1)$ @ *stage* 1: $q(1,2,1) - \min(ech(1), ecc(2))\, z_{1,2,1} \ll 0$

$\log_q(1,2,2)$ @ *stage* 2: $q(1,2,2) - \min(ech(1), ecc(2))\, z_{1,2,2} \ll 0$

$\log_q(2,1,1)$ @ *stage* 1: $q(2,1,1) - \min(ech(2), ecc(1))\, z_{2,1,1} \ll 0$

$\log_q(2,1,2)$ @ *stage* 2: $q(2,1,2) - \min(ech(2), ecc(1))\, z_{2,1,2} \ll 0$

$\log_q(2,2,1)$ @ *stage* 1: $q(2,2,1) - \min(ech(2), ecc(2))\, z_{2,2,1} \ll 0$

$\log_q(2,2,2)$ @ *stage* 2: $q(2,2,2) - \min(ech(2), ecc(2))\, z_{2,2,2} \ll 0$

$\log_{qc}(1): q_c(1) - ech\ (1) z_{cu}(1) \ll 0$

$\log_{qc}(2): q_c(2) - ech\ (2) z_{cu}(2) \ll 0$

$\log_{qh}(1): q_h(1) - ecc\ (1) z_{hu}(1) \ll 0$

$\log_{qh}(2): q_h(2) - ecc\ (2) z_{hu}(2) \ll 0$

EXPLANATION ON THE CODE (2)



$\log_{dth}(1,1,1)$ @ *stage* 1: $\Delta t(1,1,1) \ll th(1,1) - tc(1,1) + \gamma(1,1)(1 - z(1,1,1))$

$\log_{dth}(1,2,1)$ @ *stage* 1: $\Delta t(1,2,1) \ll th(1,1) - tc(2,1) + \gamma(1,2)(1 - z(1,2,1))$

$\log_{dth}(2,1,1)$ @ *stage* 1: $\Delta t(2,1,1) \ll th(2,1) - tc(1,1) + \gamma(2,1)(1 - z(2,1,1))$

$\log_{dth}(2,2,1)$ @ *stage* 1: $\Delta t(2,2,1) \ll th(2,1) - tc(2,1) + \gamma(2,2)(1 - z(2,2,1))$

$\log_{dth}(1,1,2)$ @ *stage* 2: $\Delta t(1,1,2) \ll th(1,2) - tc(1,2) + \gamma(1,1)(1 - z(1,1,2))$

$\log_{dth}(1,2,2)$ @ *stage* 2: $\Delta t(1,2,2) \ll th(1,2) - tc(2,2) + \gamma(1,2)(1 - z(1,2,2))$

$\log_{dth}(2,1,2)$ @ *stage* 2: $\Delta t(2,1,2) \ll th(2,2) - tc(1,2) + \gamma(2,1)(1 - z(2,1,2))$

$\log_{dth}(2,2,2)$ @ *stage* 2: $\Delta t(2,2,2) \ll th(2,2) - tc(2,2) + \gamma(2,2)(1 - z(2,2,2))$

$\log_{dtc}(1,1,1)$ @ *stage* 1: $\Delta t(1,1,2) \ll th(1,2) - tc(1,2) + \gamma(1,1)(1 - z(1,1,1))$

Explanation on the code (2)



Upper bound driving force

$$\gamma(1,1) = \max\left(0, \overbrace{tcin(1) - thin(1)}^{\ll 0}, \overbrace{tcin(1) - thout(1)}^{\ll 0}, \overbrace{tcout(1) - thin(1)}^{\ll 0}, \overbrace{tcout(1) - thout(1)}^{\ll 0 \text{ or } \gg 0}\right)$$



$$if\ z(i,1,1) = 0 \rightarrow \gamma(1,1) \rightarrow 0/\gg 0\ (not\ relevant.\ no\ match)$$

$$z(1,1,1) = 1 \rightarrow \Delta t(1,1,1) \ll th(1,1) - tc(1,1) \rightarrow hot\ end$$

$$\rightarrow \Delta t(1,1,2) \ll th(1,2) - tc(1,2) \rightarrow cold\ end$$

Chemical Engineering
Inspiring Potential·Generating Futures

UNIVERSITI TEKNOLOGI PETRONAS

EXPLANATION ON THE CODE (2)



$$\log_{dthu}(1,1)\ @\ first\ interval: \Delta thu(1) \ll thuout - tc(1,1)$$

$$\log_{dthu}(2,1)\ @\ first\ interval: \Delta thu(2) \ll thuout - tc(2,1)$$

$$If\ thuout = 370K; tc(1,1) = 300K; tcout(1) = 350K$$

$$\log_{dtcu}(1,3)\ @\ last\ interval: \Delta tcu(1) \ll th(1,3) - tcuout$$

$$\log_{dtcu}(2,3)\ @\ last\ interval: \Delta tcu(2) \ll th(2,3) - tcuout$$

EXPLANATION ON THE CODE (2)



$$obj: cost = 5500\left(\left(z_{(1,1,1)} + z_{(1,1,2)} + z_{(1,2,1)} + z_{(1,2,2)} \cdots z_{(2,2,2)}\right) + zcu(1) + zcu(2) + zhu(1) + zhu(2)\right)$$

$$+ 150 \frac{q(1,1,1)\left(\frac{1}{1} + \frac{1}{1}\right)}{\Delta t\ (1,1,1) \cdot \Delta t(1,1,2)\left(\frac{\Delta t(1,1,1) + \Delta t(1,1,2)}{2}\right) + (10^{-6})^{0.33}}$$

RESULTS COMPARISON



a). Automated results using two solvers (Baron and Dicopt)



b). Hand calculation

Chemical Engineering
Inspiring Potential·Generating Futures

AUTOMATED RESULT USING 3 STAGES



Design HEN for your case studies using GAMS

Simplify the design

COURSE OVERVIEW

UNIVERSITI TEKNOLOGI PETRONAS



Chemical Engineering
Inspiring Potential · Generating Futures